The Mathematica® Journal

Evaluation of Gaussian Molecular Integrals

III. Nuclear-Electron Attraction Integrals

Minhhuy Hô
Julio Manuel Hernández-Pérez

This article carries out the evaluation of nuclear-electron attraction energy integrals using Gaussian-type functions with arbitrary Cartesian angular values. As an example, we calculate the corresponding matrix for the water molecule in the STO-3G basis set.

Introduction

Evaluating molecular integrals has been an active field since the middle of the last decade. Efficient algorithms have been developed and implemented in various programs. Detailed accounts of molecular integrals can be found in the references of [1]. In this article, the third in a series describing algorithms for evaluating molecular integrals, we detail the evaluation of the nuclear-electron attraction energy integrals from a more didactic point of view, following the approach of Rys, Dupuis, and King [2] as implemented in the OpenMol program [3].

The energy caused by the attraction between an electron in the region described by the overlap of the orbitals χ_a, χ_b and a nuclear of charge Z located at $\mathbf{R}$ is expressed by the nuclear-electron attraction integral

$$\text{NE} = -Z \int_{-\infty}^{\infty} \chi_a(\mathbf{r};\, \alpha, \mathbf{A}, \mathbf{a}) \left(\frac{1}{|\, \mathbf{r} - \mathbf{R}\,|} \right) \chi_b(\mathbf{r};\, \beta, \mathbf{B}, \mathbf{b})\, d\mathbf{r}, \tag{1}$$

in which $\chi_a(\mathbf{r})$ is an unnormalized Cartesian Gaussian primitive.

Using the Gaussian product (see, for example [1]) and defining the angular part $(x - A_x)^{a_x} (y - B_x)^{b_x} \left(y - A_y\right)^{a_y} \left(y - B_y\right)^{b_y} (z - A_z)^{a_z} (z - B_z)^{b_z}$ as Θ:

$$\chi_a(\mathbf{r};\, \alpha, \mathbf{A}, \mathbf{a}) \cdot \chi_b(\mathbf{r};\, \beta, \mathbf{B}, \mathbf{b}) = E_{AB} \times \Theta \times \exp\left[-p\, |\, r - P\,|^2\right]. \tag{2}$$

The pole problem $1/|\mathbf{r}-\mathbf{R}|$ can be solved by the Laplace transform

$$\frac{1}{|\mathbf{r}-\mathbf{R}|} = \frac{2}{\sqrt{\pi}} \int_0^\infty \exp\left[-u^2 \left|\mathbf{r}-\mathbf{R}\right|^2\right] du, \tag{3}$$

which turns the NE integral into

$$\text{NE} = E_{AB} \frac{2}{\sqrt{\pi}} \int_0^\infty \int_{-\infty}^\infty \Theta \times \exp\left[-p \left|r-P\right|^2\right] \exp\left[-u^2 \left|\mathbf{r}-\mathbf{R}\right|^2\right] d\mathbf{r}\, du, \tag{4}$$

where for now, we have ignored the factor $-Z$. In the following steps, we will make certain modifications, knowing in advance that they will help simplify the expressions later on. We first reduce the upper limit of u to unity by making the changes of variable (recall from the Gaussian product that $p = \alpha + \beta$):

$$\begin{aligned} t^2 &= \frac{u^2}{p+u^2}, \\ u^2 &= \frac{t^2\, p}{1-t^2}, \end{aligned} \tag{5}$$

and

$$\begin{aligned} dt &= \frac{\left(1-t^2\right)^{3/2}}{\sqrt{p}}\, du, \\ du &= \frac{\sqrt{p}}{\left(1-t^2\right)^{3/2}}\, dt. \end{aligned} \tag{6}$$

Replace u, u^2, and du in NE, to get

$$\begin{aligned} \text{NE} = E_{AB}\, 2 \sqrt{\frac{p}{\pi}} \\ \int_0^1 \frac{1}{\left(1-t^2\right)^{3/2}} \int_{-\infty}^\infty \Theta \times \exp\left[-p \left|r-P\right|^2\right] \exp\left[-\frac{t^2\, p}{1-t^2} \left|\mathbf{r}-\mathbf{R}\right|^2\right] d\mathbf{r}\, dt. \end{aligned} \tag{7}$$

We now multiply NE by the factor $1 = (\pi/p)/(p/\pi)$:

$$\begin{aligned} \text{NE} = E_{AB} \frac{2\pi}{p} \int_0^1 \left(\frac{p}{\pi}\right)^{3/2} \frac{1}{\left(1-t^2\right)^{3/2}} \\ \int_{-\infty}^\infty \Theta \times \exp\left[-p \left|r-P\right|^2\right] \exp\left[-\frac{t^2\, p}{1-t^2} \left|\mathbf{r}-\mathbf{R}\right|^2\right] d\mathbf{r}\, dt. \end{aligned} \tag{8}$$

Again, by inserting $1 = \exp\left[-p\,t^2 \left|\mathbf{P}-\mathbf{R}\right|^2\right] \exp\left[p\,t^2 \left|\mathbf{P}-\mathbf{R}\right|^2\right]$, we get

$$\begin{aligned}\text{NE} = E_{AB}\,\frac{2\,\pi}{p}\int_0^1 \left(\frac{p}{\pi}\right)^{3/2} \frac{1}{\left(1-t^2\right)^{3/2}} \exp\left[-p\,t^2 \left|\mathbf{P}-\mathbf{R}\right|^2\right] \\ \left\{\int_{-\infty}^{\infty} \Theta \times \exp\left[-p\,|\,r-R\,|^2\right] \times \exp\left[-\frac{t^2\,p}{1-t^2}\left|\mathbf{r}-\mathbf{R}\right|^2\right] d\mathbf{r}\right\} dt.\end{aligned} \tag{9}$$

Having arrived at the desired form, we reinsert the value of the angular part Θ into the expression and separate the term enclosed by the curly brackets into three components x, y, and z:

$$\begin{aligned}\text{NE} = E_{AB}\,\frac{2\,\pi}{p}\int_0^1 \left(\frac{p}{\pi}\right)^{3/2} \frac{1}{\left(1-t^2\right)^{3/2}} \\ \left\{\exp\left[-\text{pt}^2(P_x-R_x)^2\right] \int_{-\infty}^{\infty} (x-A_x)^{a_x}\,(x-B_x)^{b_x} \exp\left[-p(x-P_x)^2\right] \times \right. \\ \exp\left[-\frac{t^2\,p}{1-t^2}\,(x-R_x)^2\right] dx \exp\left[-\text{pt}^2\left(P_y-R_y\right)^2\right] \int_{-\infty}^{\infty} \left(y-A_y\right)^{a_y} \\ \left(y-B_y\right)^{b_y} \exp\left[-p\left(y-P_y\right)^2\right] \times \exp\left[-\frac{t^2\,p}{1-t^2}\left(y-R_y\right)^2\right] \\ dy \exp\left[-\text{pt}^2(P_z-R_z)^2\right] \int_{-\infty}^{\infty} (z-A_z)^{a_z}\,(z-B_z)^{b_z} \\ \left.\exp\left[-p(z-P_z)^2\right] \times \exp\left[-\frac{t^2\,p}{1-t^2}\,(z-R_z)^2\right] dz\right\} dt.\end{aligned} \tag{10}$$

Defining $n_x(a_x, b_x; t)$ as the function of the x component inside the bracket,

$$\begin{aligned}n_x(a_x, b_x; t) = \exp\left[-p\,t^2(P_x-R_x)^2\right] \\ \int_{-\infty}^{\infty} (x-A_x)^{a_x}\,(x-B_x)^{b_x} \exp\left[-p(x-P_x)^2\right] \times \exp\left[-\frac{t^2\,p}{1-t^2}\,(x-R_x)^2\right] dx,\end{aligned} \tag{11}$$

and similarly for n_y and n_z, we rewrite the NE integral as

$$\begin{aligned}\text{NE} = E_{AB}\,\frac{2\,\pi}{p}\int_0^1 \left(\frac{p}{\pi}\right)^{3/2} \frac{1}{\left(1-t^2\right)^{3/2}} \\ \exp\left[-p\,t^2 \left|\mathbf{P}-\mathbf{R}\right|^2\right] \left\{n_x(a_x, b_x; t)\, n_y\left(a_y, b_y; t\right) n_z(a_z, b_z; t)\right\} dt.\end{aligned} \tag{12}$$

We will show that the integrand in the expression for $n_x(a_x, b_x; t)$ is in fact an overlap between two one-dimensional Gaussians, and we may use the results that have been developed in [1]. First, we expand the exponential parts of the integrand

$$\exp\Big[-p(x-P_x)^2\Big]\times\exp\Big[-\frac{t^2\,p}{1-t^2}\,(x-R_x)^2\Big]= \\ \exp\Big[-p\big(x^2-2\,\mathrm{xP}_x+{P_x}^2\big)-\frac{t^2\,p}{1-t^2}\,\big(x^2-2\,x\,R_x+{R_x}^2\big)\Big]; \tag{13}$$

regrouping in terms of p and $\big(t^2\,p\big)\big/\big(1-t^2\big)$, we have

$$\exp\Big[-\left(p+\frac{t^2\,p}{1-t^2}\right)x^2+2\left(p\,P_x+\frac{t^2\,p}{1-t^2}\,R_x\right)x-p\,{P_x}^2-\frac{t^2\,p}{1-t^2}\,{R_x}^2\Big], \tag{14}$$

which becomes

$$\exp\Big[-(\lambda+\sigma)\,x^2+2\,(\lambda\,P_x+\sigma\,P_x)\,x-\lambda\,{P_x}^2-\sigma\,{R_x}^2\Big], \tag{15}$$

where $\lambda = p$ and $\sigma = \big(t^2\,p\big)\big/\big(1-t^2\big)$. These definitions let us compare this equation with the result of the Appendix, in which we see that equation (15) is simply

$$\exp\Big[-(\lambda+\sigma)\,(x-Q_x)^2\Big]\exp\Big[-\frac{\lambda\,\sigma(P_x-R_x)^2}{\lambda+\sigma}\Big], \tag{16}$$

where

$$Q_x=\frac{\lambda\,P_x-\sigma\,R_x}{\lambda+\sigma}. \tag{17}$$

Substituting $\lambda+\sigma$ and Q_x,

$$\lambda+\sigma=p+\frac{t^2\,p}{1-t^2}=\frac{p}{1-t^2}, \\ Q_x=\frac{\lambda\,P_x+\sigma\,R_x}{\lambda+\sigma}=P_x-t^2(P_x-R_x), \tag{18}$$

into equation (13), we have

$$\exp\Big[-p(x-P_x)^2\Big]\exp\Big[-\frac{t^2\,p}{1-t^2}\,(x-R_x)^2\Big]= \\ \exp\Big[-\frac{p}{1-t^2}\,\big(x-\big[P_x-t^2(P_x-R_x)\big]\big)^2\Big]\exp\Big[-\frac{p^2\,t^2}{1-t^2}\,\frac{(P_x-R_x)^2}{p\big/\big(1-t^2\big)}\Big]. \tag{19}$$

Substitute this result into the definition of $n_x(a_x, b_x; t)$ to get

$$n_x(a_x, b_x; t) = \left(\frac{p}{(1-t^2)\,\pi}\right)^{1/2} \int_{-\infty}^{\infty} (x - A_x)^{a_x}\,(x - B_x)^{b_x} \exp\left[-\frac{p}{1-t^2}\left(x - \left[P_x - t^2(P_x - R_x)\right]\right)^2\right] dx. \tag{20}$$

The integral has the same form as a one-dimensional overlap integral where the integrand is a Gaussian function centered at $P_x - t^2(P_x - R_x)$ with an exponential coefficient $p/(1-t^2)$.

Recurrence Relations

From the observation above, we make use of the results developed for overlap integrals in [1]. For example, for $a_x = b_x = 0$,

$$n_x(0, 0; t) = 1. \tag{21}$$

In particular, we have the transfer equations

$$\begin{aligned} &n_x(a_x + 1, 0; t) = \\ &\quad -(A_x - P_x + t(P_x - R_x))\, n_x(a_x; 0; t) + \frac{a_x}{2\,p}\left(1 - t^2\right) n_x(a_x - 1; 0; t), \\ &n_x(a_x, b_x + 1; t) = n_x(a_x + 1, b_x; t) + (A_x - B_x)\, n_x(a_x, b_x; t). \end{aligned} \tag{22}$$

The n_y and n_z functions take similar forms. The product $n_x\, n_y\, n_z$ is a polynomial in t^2, and if we replace $T = p\,(\mathbf{P} - \mathbf{R})^2$, then the NE integral in equation (12) is

$$\text{NE} = E_{AB}\,\frac{2\,\pi}{p}\int_0^1 P\left(t^2\right) \exp\left(-T\,t^2\right) dt, \tag{23}$$

where $P\left(t^2\right)$ is the said polynomial. The integral is a combination of the Boys function (see, for example, Reference 4 of [1])

$$F_n(T) = \int_0^1 t^{2\,n} \exp\left(-T\,t^2\right) dt, \tag{24}$$

a strictly positive and decreasing function.

Implementation

Aside from the obvious choice of using *Mathematica* to evaluate the Boys function, there are several ways of evaluating the integral. In practice, most programs store pretabulated values of the function at different intervals and interpolation is done as needed (e.g. by Chebyshev polynomials). Here we use the Gauss–Chebyshev quadrature numerical integration [4]. For simplicity, we have adopted almost verbatim the F77 code in [4, p. 46].

```
abscissa[n_, i_] :=
 (n + 1 - 2 * i) / (n + 1) +
  (2 / Pi) (1 + (2 / 3) Sin[i Pi / (n + 1)] ^ 2) Cos[i Pi / (n + 1)]
   Sin[i Pi / (n + 1)]

omega[n_, i_] := (16 / (3 (n + 1))) Sin[i Pi / (n + 1)] ^ 4;

IntChebyshev[eps_, M_, F_] :=
 Module[{c0 = Cos[Pi / 6.], s0 = Sin[Pi / 6.], c1, s1, q,
   p, chp, j, c, s, xp, err = 10., n = 3},
  c1 = s0;
  s1 = c0;
  q = ((F /. x → abscissa[2, 1]) + (F /. x → -abscissa[2, 1])) *
    omega[2, 1];
  p = F /. x → 0.0;
  chp = q + p;
  j = 0;
  While[
   (err > eps) && ((2 n (1 - j) + j * 4 * n / 3 - 1) ≤ M),

   j = 1 - j;
   c1 = j * c1 + (1 - j) * c0;
   s1 = j * s1 + (1 - j) * s0;
   c0 = j * c0 + (1 - j) * Sqrt[(1 + c0) * 0.5];
   s0 = j * s0 + (1 - j) * s0 / (c0 + c0);
   c = c0;
   s = s0;
   Do[
    xp = 1 + 2 / (3 * Pi) * s * c * (3 + 2 * s * s) - i / n;
    If[
     Ceiling[(3 * (i + j + j) / 3)] > i + j,
     chp = chp + (((F /. x → -xp) + (F /. x → xp)) * s ^ 4)
    ]; (* end of If *)

    xp = s;
    s = s * c1 + c * s1;
    c = c * c1 - xp * s1,
    {i, 1, n - 1, 2}
   ]; (* End of Do Loop *)
```

```
  n = (1 + j) * n;
  p = p + (1 - j) * (chp - q);
  err = 16 * Abs[(1 - j) * (q - 3 * p / 2) + j * (chp - 2 * q)] / (3 n);
  q = (1 - j) * q + j * chp
 ]; (* end of While *)

 SetPrecision[16 * q / (3 n), 16]
]
```

The function `Nea` evaluates the nuclear-electron attraction integral of two Gaussian primitives; here `alpha`, `beta`, `RA`, `RB`, `LA`, and `LB` are α, β, $\mathbf{A}$, $\mathbf{B}$, $\mathbf{a}$, and $\mathbf{b}$ as defined earlier; `RR` is the nuclear position.

```
Nea[alpha_, beta_, RA_, RB_, RR_, LA_, LB_] := Module[
 {eta, EAB},
 Do[
  (* Initial Conditions for eta *)

  eta[i, 0, 0] = 1;
  eta[i, 1, 0] =
   -(RA[[i]] - ((alpha * RA[[i]] + beta * RB[[i]]) /
         (alpha + beta)) +
      (t^2) *
       (((alpha * RA[[i]] + beta * RB[[i]]) / (alpha + beta)) -
         RR[[i]]));

  (* Recurrence Index *)

  eta[i_, a_, 0] :=
   -(RA[[i]] - ((alpha * RA[[i]] + beta * RB[[i]]) /
           (alpha + beta)) +
        (t^2) *
         (((alpha * RA[[i]] + beta * RB[[i]]) / (alpha + beta)) -
           RR[[i]])) * eta[i, a - 1, 0] +
    ((a - 1) / (2 * (alpha + beta))) * (1 - t^2) * eta[i, a - 2, 0];

  (* Transfer Equation *)

  eta[i_, a_, b_] :=
   eta[i, a + 1, b - 1] + (RA[[i]] - RB[[i]]) * eta[i, a, b - 1];,
  {i, 1, 3}];

 (* Exponential Overlap *)

 EAB = Exp[-(alpha * beta / (alpha + beta)) *
    (RA - RB).(RA - RB)];

 (* N-e Attraction Energy Integral *)
```

```
EAB * (2 * Pi / (alpha + beta)) *

 (* Numerical integration using Pérez-Jordá method *)
 (* 10^-10 tolerance, 50000 maximum points allowed *)
 (* Equation (23) *)

 (IntChebyshev[10^-10, 50 000,

   1/2 *
    (Exp[-((alpha + beta) * (t^2) *
           ((((alpha * RA + beta * RB) / (alpha + beta)) - RR).
             (((alpha * RA + beta * RB) / (alpha + beta)) -
               RR)))] * eta[1, LA[[1]], LB[[1]]] *
       eta[2, LA[[2]], LB[[2]]] *
       eta[3, LA[[3]], LB[[3]]] /. t → (x + 1) / 2)])]
```

As in our two earlier articles [1, 5], we use the same data for the water molecule (r_{OH} = 1.86942 bohr, $\angle_{HOH}$ = 100.0269 °, the geometry optimized at the HF/STO-3G level). The molecule lies in the *y*-*z* plane with Cartesian coordinates in atomic units.

```
R = {
    {0., 1.43233673, -0.96104039},
    {0., -1.43233673, -0.96104039},
    {0., 0., 0.24026010}
   };
```

In the STO-3G basis set, each atomic orbital is approximated by a sum of three Gaussians; here are their unnormalized primitive contraction coefficients and orbital exponents.

```
PrimCoeff = {
    {0.1543289673, 0.5353281423, 0.4446345422},
    {0.1543289673, 0.5353281423, 0.4446345422},
    {0.1543289673, 0.5353281423, 0.4446345422},
    {-0.09996722919, 0.3995128261, 0.7001154689},
    {0.155916275, 0.6076837186, 0.3919573931},
    {0.155916275, 0.6076837186, 0.3919573931},
    {0.155916275, 0.6076837186, 0.3919573931}
   };

OrbCoeff = {
    {3.425250914, 0.6239137298, 0.168855404},
    {3.425250914, 0.6239137298, 0.168855404},
    {130.7093214, 23.80886605, 6.443608313},
```

```
    {5.033151319, 1.169596125, 0.38038896},
    {5.033151319, 1.169596125, 0.38038896},
    {5.033151319, 1.169596125, 0.38038896},
    {5.033151319, 1.169596125, 0.38038896}
   };
```

Here are the basis function origins and Cartesian angular values of the orbitals, listed in the order $H_1(1\,s)$, $H_2(1\,s)$, $O_3(1\,s)$, $O_3(2\,s)$, $O_3(2p_x)$, $O_3(2p_y)$, and $O_3(2p_z)$.

```
FCenter = {R[[1]], R[[2]], R[[3]], R[[3]], R[[3]],
    R[[3]], R[[3]]};

CartAng = {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}, {0, 0, 0},
    {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};

Z = {1, 1, 8};
```

Specifically, for the nuclear-electron attraction energy integral between the first primitive of the $1\,s$ orbital of hydrogen atom 1, $\chi_{11}(\mathbf{r})$, the first primitive of the $2\,p_z$ orbital of the oxygen atom, $\chi_{71}(\mathbf{r})$, and atom 1 ($Z = 1$) is

$$-\int_{-\infty}^{\infty}\chi_{11}(\mathbf{r})\left(\frac{1}{\mathbf{r}-\mathbf{R_1}}\right)\chi_{71}(\mathbf{r})\,d\mathbf{r} = \int_{-\infty}^{\infty}\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} e^{-3.425250914\,|\mathbf{r}-\mathbf{R}_{H1}|^2}\left(\frac{1}{\mathbf{r}-\mathbf{R_1}}\right)(z-0.24026010)\times e^{-5.033151319\,|\mathbf{r}-\mathbf{R}_{O3}|^2}\,dx\,dy\,dz. \tag{25}$$

We have

$$E_{AB}\,\frac{2\pi}{\alpha+\beta} = 0.000599054,\ \text{eta}(x,0,0) = 1.0,\ \text{eta}(y,0,0) = 1.0,$$
$$\text{eta}(z,0,1) = \text{eta}(z,1,0) + \mathbf{R}_{H1(z)} + \mathbf{R}_{O3(z)} = -0.4867 - 0.71843\,t^2.$$

From the Gauss–Chebyshev quadrature, the integral in equation (23) yields $\int_0^1\left(-0.4867-0.71843\,t^2\right)\exp\left[-p\,(\mathbf{P}-\mathbf{R})^2\,t^2\right]dt = -0.1426225$. The nuclear-electron integral (25) is $0.000599054\times(-0.1426225) = -0.0000854386$. This is calculated as follows.

```
Nea[OrbCoeff[[1, 1]], OrbCoeff[[7, 1]], R[[1]], R[[3]],
 R[[1]], CartAng[[1]], CartAng[[7]]]

-0.0000854386
```

We would first need the normalization factor before evaluating the nuclear-electron energy matrix.

```
NormCoeff[α_, {ax_, ay_, az_}] :=
  (2 α/π)^(3/4) (4 α)^(ax+ay+az)/2 / ((2 ax - 1)!! (2 ay - 1)!! (2 az - 1)!!)^(1/2);

Table[Sum[(-Z[[Nuc]]) * Sum[
      NormCoeff[OrbCoeff[[p, i]], CartAng[[p]]]
       * NormCoeff[OrbCoeff[[q, j]], CartAng[[q]]]
       * PrimCoeff[[p, i]]
       * PrimCoeff[[q, j]]
       * Nea[
        OrbCoeff[[p, i]], OrbCoeff[[q, j]],
        FCenter[[p]], FCenter[[q]], R[[Nuc]],
        CartAng[[p]], CartAng[[q]]
       ],
      {i, 3}, {j, 3}], {Nuc, 1, 3}], {p, 1, 7}, {q, 1, 7}
 ] // MatrixForm // Chop
```

```
⎛-5.71691  -1.56861  -1.61524  -3.65729      0      -2.09047    1
⎜-1.56861  -5.71691  -1.61524  -3.65729      0       2.09047    1
⎜-1.61524  -1.61524  -61.6912  -7.43668      0          0      0.(
⎜-3.65729  -3.65729  -7.43668  -10.1107      0          0      0.
⎜    0          0         0         0   -9.96004      0
⎜-2.09047   2.09047      0         0          0     -10.0963
⎝  1.826     1.826   0.0186812 0.222158      0          0      -1
```

Conclusion

We have provided a didactic introduction to the evaluation of nuclear-electron attraction-energy integrals involving Gaussian-type basis functions by use of recurrence relations and a numerical quadrature scheme. The results are sufficiently general so that no modification of the algorithm is needed when larger basis sets with more Gaussian primitives or primitives with larger angular momenta are employed.

Appendix

Consider the Gaussian product: $\exp\left[-\alpha\,(\mathbf{r}-\boldsymbol{A})^2\right]\exp\left[-\beta\,(\mathbf{r}-\mathbf{B})^2\right]$. Combine and expand the coefficients to get

$$\begin{aligned}
&\exp\left[-\alpha\,(\mathbf{r}-\boldsymbol{A})^2\right]\exp\left[-\beta\,(\mathbf{r}-\mathbf{B})^2\right]\\
&= \exp\left[-\left(\alpha\,\mathbf{r}^2 - 2\,\alpha\,\mathbf{r}\,\boldsymbol{A} + \alpha\,\boldsymbol{A}^2 + \beta\,\mathbf{r}^2 - 2\,\beta\,\mathbf{r}\,\mathbf{B} + \beta\,\mathbf{B}^2\right)\right]\\
&= \exp\left[-\left((\alpha+\beta)\,\mathbf{r}^2 - 2\,(\alpha\,\boldsymbol{A}+\beta\,\mathbf{B})\,\mathbf{r} + \alpha\,\boldsymbol{A}^2 + \beta\,\mathbf{B}^2\right)\right]\\
&= \exp\left[-(\alpha+\beta)\left(\mathbf{r}^2 - \frac{2\,(\alpha\,\boldsymbol{A}+\beta\,\mathbf{B})}{(\alpha+\beta)}\,\mathbf{r} + \frac{\alpha\,\boldsymbol{A}^2+\beta\,\mathbf{B}^2}{(\alpha+\beta)}\right)\right].
\end{aligned}$$

Let $\mathbf{P} = \frac{(\alpha\,\boldsymbol{A}+\beta\,\mathbf{B})}{(\alpha+\beta)}$ and substitute $\mathbf{P}^2 - \mathbf{P}^2 = \mathbf{P}^2 - \frac{(\alpha\,\boldsymbol{A}+\beta\,\mathbf{B})^2}{(\alpha+\beta)^2}$ in the exponent to get

$$\begin{aligned}
&\exp\left[-\alpha\,(\mathbf{r}-\boldsymbol{A})^2\right]\exp\left[-\beta\,(\mathbf{r}-\mathbf{B})^2\right] =\\
&\exp\left[-(\alpha+\beta)\left(\mathbf{r}^2 - 2\,\mathbf{P}\,\mathbf{r} + \mathbf{P}^2 - \frac{(\alpha\,\boldsymbol{A}+\beta\,\mathbf{B})^2}{(\alpha+\beta)^2} + \frac{\alpha\,\boldsymbol{A}^2+\beta\,\mathbf{B}^2}{(\alpha+\beta)}\right)\right].
\end{aligned}$$

The first three terms inside the second bracket factor to $(\mathbf{r}-\mathbf{P})^2$, and the last two can be reduced to $\frac{\alpha\,\beta\,(\mathbf{A}-\mathbf{B})^2}{(\alpha+\beta)^2}$. The original Gaussian product is thus

$$\exp\left[-\alpha\,(\mathbf{r}-\boldsymbol{A})^2\right]\exp\left[-\beta\,(\mathbf{r}-\mathbf{B})^2\right] = \exp\left[-(\alpha+\beta)\,(\mathbf{r}-\mathbf{P})^2\right]\exp\left[-\frac{\alpha\,\beta\,(\mathbf{A}-\mathbf{B})^2}{(\alpha+\beta)}\right].$$

Here is a verification.

```
Module[
 {r, x, y, z, A, ax, ay, az, B, bx, by, bz, P, α, β},
 r = {x, y, z};
 A = {ax, ay, ax};
 B = {bx, by, bz};
 P = (α A + β B) / (α + β);
 Exp[-α (r - A).(r - A)] Exp[-β (r - B).(r - B)] ==
   Exp[-(α + β) (r - P).(r - P)]
    Exp[-(α β (A - B).(A - B)) / (α + β)] // Simplify
]

True
```

If one opts to use *Mathematica*'s own incomplete gamma function $\Gamma(a, z)$, equation (24) can be defined in closed form.

```
F[u_, n_] := 1/2 u^(-n-1/2) (Gamma[n + 1 / 2] - Gamma[n + 1 / 2, u])
```

Here is an example.

```
example = Exp[-α t^2] Sum[β_k t^(2 k), {k, 0, 6}] /.
  {α → 3, β_k_ :→ 1 / (k + 1`10)}
```

```
e^(-3 t^2) (1.000000000 + 0.5000000000 t^2 +
    0.3333333333 t^4 + 0.25000000000 t^6 + 0.20000000000 t^8 +
    0.16666666667 t^10 + 0.14285714286 t^12)
```

This is the result.

```
Expand[t^k example] /. t^(k+n_.) Exp[a_ t^2] :→ F[-a, n / 2]
```

```
0.5605179384
```

Here is a comparison with the quadrature result.

```
IntChebyshev[10^-10, 50 000, 1/2 example /. t → (x + 1)/2]
```

```
0.5605451373144739
```

Understandably, direct use of the `Gamma` function is much faster and yields more accurate results. We thank Paul Abbott for providing the example.

References

[1] M. Hô and J. M. Hernández-Pérez, "Evaluation of Gaussian Molecular Integrals I," *The Mathematica Journal*, **14**(3), 2012. doi:10.3888/tmj.14–3.

[2] J. Rys, M. Dupuis, and H. F. King, "Computation of Electron Repulsion Integrals Using the Rys Quadrature Method," *Journal of Computational Chemistry*, **4**(2), 1983 pp. 154–157. doi:10.1002/jcc.540040206.

[3] G. H. F. Diercksen and G. G. Hall, "Intelligent Software: The OpenMol Program," *Computers in Physics*, **8**(2), 1994 pp. 215–222. doi:10.1063/1.168520.

[4] J. Pérez-Jorda and E. San-Fabián, "A Simple, Efficient and More Reliable Scheme for Automatic Numerical Integration," *Computer Physics Communications*, **77**(1), 1993 pp. 46–56. doi:10.1016/0010-4655(93)90035-B.

[5] M. Hô and J. M. Hernández-Pérez, "Evaluation of Gaussian Molecular Integrals II," *The Mathematica Journal*, **15**(1), 2013. doi:10.3888/tmj.15–1.

M. Hô and J. M. Hernández-Pérez, "Evaluation of Gaussian Molecular Integrals," *The Mathematica Journal*, 2014. dx.doi.org/doi:10.3888/tmj.16-9.

About the Authors

Minhhuy Hô received his Ph.D. in theoretical chemistry at Queen's University, Kingston, Ontario, Canada, in 1998. He is currently a professor at the Centro de Investigaciones Químicas at the Universidad Autónoma del Estado de Morelos in Cuernavaca, Morelos, México.

Julio-Manuel Hernández-Pérez obtained his Ph.D. at the Universidad Autónoma del Estado de Morelos in 2008. He has been a professor of chemistry at the Facultad de Ciencias Químicas at the Benemérita Universidad Autónoma de Puebla since 2010.

Minhhuy Hô
Universidad Autónoma del Estado de Morelos
Centro de Investigaciones Químicas
Ave. Universidad, No. 1001, Col. Chamilpa
Cuernavaca, Morelos, Mexico CP 92010
homh@uaem.mx

Julio-Manuel Hernández-Pérez
Benemérita Universidad Autónoma de Puebla
Facultad de Ciencias Químicas
Ciudad Universitaria, Col. San Manuel
Puebla, Puebla, Mexico CP 72570
julio.hernandez@correo.buap.mx